
News  
 
O.K. the first newsletter. ANY comments about format or whether on some emailers the images are not 
visible would be very helpful. I�m feeling my way with this!  
 
I would like to also put in a �did you know� section to mention things like the Burst analysis script W_Burst 
for Spike2 that is shipped with Spike2 and can be found in the Spike2\scripts folder. For Signal (and 
actually for Spike2 too) ToolMake which builds a skeleton script for you. A good aid to getting started. 
These can also be found in the scripts folders.  
 
Version 5 of Spike2 is now released. A demonstration version of the software is available from 
http://www.ced.co.uk/pru.shtml  
 
Version 4.15 of Spike2 is freely downloadable for registered v4 users.  
Version 2.13 of Signal is also freely downloadable for registered v2 users.  
 
Version 5  
One of the main sections that Greg and his team have been working on in this new release is the 
clustering of spike data. This allows you to �cluster cut� using ellipses and write back the information to 
form templates to be used on or off-line. They have also been working on spreadsheet format output. This 
resamples the data to allow channels possibly at different original sample rates to be placed in columns 
for import directly into spreadsheets.  
 
User group. See end of newsletter.  
 
UK training days in Cambridge on Monday June 16th and Tuesday 17th. There will also be the usual 
USN training days later in the year.  
 
 
SPIKE2  
 
Q. Rate displays always start at zero. How can I change that? 
  
A.  There is already a rate display option in Spike2 (view - channel draw mode). This may not be 

quite what we need though as the first bin of the rate always starts at zero time. By using the 
active cursor (0) and the XY view from the cursor pull down we can get a plot of time against sum 

#1 ~ May 2003 

http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/nw9u.htm
http://www.ced.co.uk/nw9y.htm


of the events in a given range and also to start the plot at virtually any time in the file.  
 
Use Analysis - measurements -XY view to plot on the X 'Time at point' where the time 
is = cursor(0) on the Y 'Sum' where the start is cursor(0) and the end is Cursor(0)+5 indicating a 
5 second time window. When you press New this will produce an empty window and a process 
dialog where you can specify the start and end times to be calculated. The result is a channel of 
rate data that can now be copied as text. 
  
This method could also be used to down sample a channel if Y 'sum' was replaced with 'value at 
point' If we were only interested in a view of the rate then we could also use Analysis - Channel 
process and add Time shift in a positive or negative direction. Handy for aligning bursts of activity 
with stimulus ramps. 
  
This is a v4 or v5 function.  
 

 
 
Spike templates  
 
Q. I know that it is the same spike but over time the amplitude has changed and I get another 

template formed even though I have set the parameters to �track�.  
 
A. The spike program may not detect the perhaps sudden change in the amplitude of a spike shape. 

We can track gradual changes but on occasion it is better to use multiple templates per spike. In 
the New or Edit Wavemark dialog. Double clicking on the template �number� can change it�s code. 
If as in the case below the code is changed to match the first template then both spikes will be 
stored in the file as 01 and coloured accordingly. The other benefit of doing this is that you do not 
create a much wider amplitude template that may allow other unwanted spikes to match. You may 
find this particularly helpful if you have a short burst of 4 or 5 spikes that rapidly change their 
amplitude. An example of this may look like�..  



 
 
 
 
This is available in all Spike2 versions.  
 
 

 
 
 

Double click the highlighted number  
 
 



SIGNAL  
 
Fast triggers and fast fixed interval sweep modes  
 
During sampling with Signal, as well as the data being transferred from the 1401 interface to the host 
computer there is generally also a great deal of other information flowing in both directions, including 
details such as the state to be used for the next sweep and any changes to pulse outputs. This causes a 
delay between the end of one sweep and Signal being �armed� to wait for the next trigger resulting in a 
lower possible sweep rate than you may think is possible. For many complex recording protocols this is 
absolutely necessary, however if a simple protocol requiring high repetition rates is needed it can cause 
problems.  
 
Using �Fast triggers� sweep mode, significantly shorter intervals are possible, resulting in Signal being 
ready for the next trigger virtually instantaneously. This is done by limiting the amount of features available 
thereby stopping the need to communicate with the host computer other than for data transfer. The 
limitations of using this are that multiple states and incremental pulse outputs are disabled.  
 
For applications requiring a very fast repetition rate but which are required to be timed from within Signal 
itself �Fast fixed interval� sweep mode is available. This gives the same limitations as �Fast triggers� but 
enables the user to set the sweep length and inter sweep interval within Signal rather than relying on an 
external trigger.  
 
 
Scripts. Spike2         Right mouse click the script icon and save file to disk.

 
We've been asked recently to produce a script and sequencer which will output a digital TTL pulse when a 
waveform threshold is reached. Can be useful for pulsing when a burst of activity is detected or marking 
ECG R waves.  More information can be found on our Spike Online Script page. 
 
 
Scripts Signal         Right mouse click the script icon and save file to disk.

 
Output Signal data files .CFS as Windows .WAV format. This allows you to import the channel or channels 
into Matlab amongst other software.  More information can be found on our Signal Display Script page.   
 
Recent questions  
 
Something simple to start I think.  
 
Q. When forming averages I would like to know how many sweeps of data have been added to the 

mean display.  
 
A. From the view pull down select info and this will give you the sweep count. It is possible to reset 

this using the script and indeed reset the average at n sweeps if you want.  
 
User group  
 
It will take a short time to set up an archive message board on our web site. As soon as this is done I will 
use this newsletter to announce that it is available. This seems to be the most sensible route to go rather 
than direct emailing both for security and for minimising spam.  


'ThreshPulse.s2s
'Written by Steven Clifford - CED - 21/01/03
'This is a script to detect a threshold crossing on a waveform input channel
'and mark this by outputting a TTL pulse or setting a marker in a memory channel
'which can be, for example, used to display heart rate. 

'It uses the current sampling configuration which must include at least one waveform channel.
'
'Note this script is UNDER DEVELOPMENT BY CED and is included for
'illustration only. We make no warranties as to its suitability
'or otherwise. It has not been thoroughly tested.
'You should use it 'as is'

'When run, the user is asked to select a waveform input channel, the direction
'of the required threshold crossing and how to mark the crossing. A data file is then started and a 
'horizontal cursor set on the input waveform channel which the user should position 
'at a reasonable starting level for the threshold. 

'The user then presses OK and 'Sample Start' to commence sampling.

'NOTE: This script writes an output sequence into a default folder of c:\Spike4\sequence into which 
'variables are passed. If this folder does not exist the script will not work. Useers of Spike 2 version 3 
'will need to edit this directory.



var seqFile$;															'Name and path for sequencer file
var data%;																'Handle of new data file
var sTime;																'The last time we looked at the idle routine
var curtrigLev;														'Current trigger level as checked by idle routine		
var oldTrigLev;														'Previous trigger level
var trigChan%;															'Memory channel used for testing purposes
var inputChan%;														'Waveform channel monitored for threshold crossings
var setDelay%;															'Delay time in ms
var delayTicks%;														'Delay in sequencer steps to pass as variable
var trigVar;															'Level of threshold in ADC units
var threshType%;														'Pulse direction
var pulseFlag% := 1;													'Flag if TTL pulse required
var memFlag% := 1;													'Flag if memory channel marker required
var numMem%;															'Number of items imported into memory channel

seqFile$ := FilePath$(2) + "Sequence\\pulse.pls";		

ToolbarVisible(1);													'Make toolbar visible always
New%();																	'Set up new sampling window
DoToolbar();															'Do the toolbar

func New%()																'New sampling window
var ok%;
var test%;
var waveChans%[32];													'List of available waveform channels	
View(LogHandle());													'Make log view the current view
EditSelectAll();														'Select all text in log view
EditClear();															'Delete it
Window(0,80,100,100);												'Display it at the bottom of the screen
WindowVisible(1);														'Make it visbible

	if data%>0 then													'If there is already a data view open then
  if ViewKind(data%)>-1 then
  	 View(data%);														'Close it
  	 FileClose();
  endif;
endif;

test% := FileNew(0,0);
if test% <= 0 then
	Message("          Unable to sample data\nPlease check 1401 and re-run script");
	Halt;
endif;
View(test%);
ChanList(waveChans%[],1);
if waveChans%[0] = 0 then
	Message("Your sampling configuration contains no waveform channels\n          Please set up suitable channel and re-run script");
	View(test%);
	FileClose(0,-1);
	Halt;
endif;

DlgCreate("Settings");									'Dialog to set channel and delay
DlgChan(1,"Input channel",1,20,1);
'DlgInteger(2,"Delay (ms)",2,1000);
DlgList(2,"Threshold","Positive|Negative",8,20,2);
DlgText("Action",2,3);
DlgCheck(3,"TTL pulse",19,3);
DlgCheck(4,"Memory channel marker",19,4);
'ok% := DlgShow(inputChan%,setDelay%,threshType%);
ok% := DlgShow(inputChan%,threshType%,pulseFlag%,memFlag%);
View(test%);
FileClose(0,-1);
if not ok% then
	Message("Pressed cancel - Script closing");
	Halt;
endif;

delayTicks%:=(setDelay%*10-4);									'Calculate number of sequencer steps to pass as variable

SampleSequencer("");													'Clear any current output sequences
if pulseFlag% then	
	WriteSequence(threshType%);									'Write output sequence to use if TTL pulse required
endif;

data%:=FileNew(0,2);													'Open a new data file for sampling
sTime := 0;
HCursorDelete(-1);					
WindowVisible(1);
if data%<0 then Message("Unable to open new data file");Halt() endif;
'DrawMode(-1,2);														'Set event draw mode to lines
Window(0,0,100,100);													'Make data window in top bit of screen
'XRange(0,10);
HCursorNew(inputChan%,oldTrigLev);								'Set cursor on appropriate channel	
Interact("Position cursor at initial threshold level",1023);
oldTrigLev:=HCursor(1);												'Get threshold level
trigVar:=(HCursor(1)-ChanOffset(inputChan%))/(5*ChanScale(inputChan%))*32768;		'Convert to ADC units
SampleSeqVar(3,trigVar);											'Set output sequencer variable
if memFlag% then
	trigChan%:=MemChan(3);	
	DrawMode(trigChan%,2);
	ChanShow(trigChan%);
endif;
'FrontView(LogHandle());											'Bring the Log view to the front
FrontView(data%);														'Bring the data view to the front
ToolbarEnable(3,0);													'Disable "Sample stop" button
ToolbarEnable(2,1);													'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"Quit",Quit%);										'Set up toolbar buttons
ToolbarSet(2,"Sample start", Start%);
ToolbarSet(3,"Sample stop", Stop%);
ToolbarSet(4,"New file", New%);
'ToolbarSet(6,"Change delay",ChangeDelay%);
ToolbarEnable(3,0);													'Disable "Sample stop" button
'ToolbarEnable(6,0);
View(data%);
Toolbar("Press SAMPLE START to commence sampling", 1023);					'Wait here until quit is pressed
end;

func Quit%()															'If "Quit" is pressed
SampleStop();															'Stop sampling
return 0;																'leave toolbar
end;

func Start%()															'If "Start" is pressed
ToolbarSet(0,"",Idle%);												'Call Idle%() whenever there is free time
SampleStart();															'Start sampling
ToolbarEnable(4,0);													'Disable "New file" button
ToolbarEnable(3,1);													'Enable "Sample stop" button
ToolbarEnable(2,0);													'Disable "Sample start" button
ToolbarEnable(1,0);													'Disable "Quit" button
'ToolbarEnable(6,1);
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;																'Stay with toolbar
end;

func Stop%()															'If "Stop" is pressed
ToolbarClear(0);
SampleStop();															'Stop sampling
ToolbarEnable(4,1);													'Enable "New file" button
ToolbarEnable(3,0);													'Disable "Sample stop" button
ToolbarEnable(1,1);													'Enable "Quit" button
'ToolbarEnable(6,0);
ToolbarText("Press FILE NEW to capture more data");
return 1;																'Stay in toolbar
end;


Func ChangeDelay%()													'Function to change delay time
setDelay%:=Input("Delay time (ms)",setDelay%,2,1000);
delayTicks%:=(setDelay%*10-4);
SampleSeqVar(4,delayTicks%);
return 1;
end;

func Idle%()															'The Idle routine is called when PC has time
var eTime;
View(data%);
eTime:=Maxtime();

'PLACE CODE IN HERE TO ANALYSE THE DATA FILE BETWEEN sTime and eTime
curTrigLev:=HCursor(1);												'Check if threshold level has changed
if curTrigLev <> oldTrigLev then									'If so, Get new level in ADC units
	trigVar:=(curTrigLev-ChanOffset(inputChan%))/(5*ChanScale(inputChan%))*32768;
	SampleSeqVar(3,trigVar);										'Update sequencer variables		
	oldTrigLev:=curTrigLev;											'Reset previous cursor level to current level
endif;
if memFlag% then														'If memory channel markers required
	if threshType% = 0 then
		numMem% := MemImport(trigChan%,inputChan%,sTime,eTime,2,0,oldTrigLev);	'import crossing times into memory channel
	else
		numMem% := MemImport(trigChan%,inputChan%,sTime,eTime,3,0,oldTrigLev);					
	endif;
	if numMem% then
		sTime:=eTime;
	endif;
endif;

return 1;																'Stay in toolbar
end;


'This function writes an output sequnce for use with the script. 
Func WriteSequence(type%)
var seq%;
var seqSet%;
seq%:=FileOpen(seqFile$,8,1);
'Message(seq%);
View(seq%);
if type% = 0 then
	Print("            SET      0.10 1 0\n");
	Print("            VAR    V1,num=0\n");
	Print("            VAR    V2,prev=0\n");
	Print("            VAR    V3,lev\n");
	Print("            VAR    V4,del=%d\n\n",delayTicks%);
	Print("UNDER:      CHAN   num,%d\n",inputChan%);
	Print("            BGT    num,lev,pulse\n");
	Print("            JUMP   under\n\n");
	Print("OVER:       CHAN   num,%d\n",inputChan%);
	Print("            BLT    num,lev,under\n");
	Print("            JUMP   over\n\n");
	Print("PULSE:      DIGOUT [00000001]\n");
	Print("            DELAY 8\n");
	Print("            DIGOUT [00000000]\n");
	Print("            JUMP   over");
else
		Print("            SET      0.10 1 0\n");
	Print("            VAR    V1,num=0\n");
	Print("            VAR    V2,prev=0\n");
	Print("            VAR    V3,lev\n");
	Print("            VAR    V4,del=%d\n\n",delayTicks%);
	Print("OVER:       CHAN   num,%d\n",inputChan%);
	Print("            BLT    num,lev,pulse\n");
	Print("            JUMP   over\n\n");
	Print("UNDER:      CHAN   num,%d\n",inputChan%);
	Print("            BGT    num,lev,over\n");
	Print("            JUMP   under\n\n");
	Print("PULSE:      DIGOUT [00000001]\n");
	Print("            DELAY 8\n");
	Print("            DIGOUT [00000000]\n");
	Print("            JUMP   under");
endif;
'FileSaveAs("c:\\Spike4\\Sequence\\Pulse.pls");
FileSave();
FileClose();
seqSet%:=SampleSequencer(seqFile$);
'Message(seqSet%);
return 1;
end;



CED Ltd
threspul.s2s
Spike2 Pulse from threshold crossing script


'=============================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB0 0FE, UK
'=============================================================================================================

'CfsToWav1.sgs
'Copyright © Cambridge Electronic Design,  Mar 2003.

'Overview.
'
'This script converts .cfs files created for example by CED Signal software to .wav format.
'It requires Signal version 2 or higher.
'The script works on one frame of data at a time and can convert up to 4 channels of data
'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'User Guide.

'Use the Run or Load and Run commands on the Signal Script menu to start the script.  
'The main script toolbar has 8 buttons that can be clicked with the mouse or operated by a hotkey (underlined).

'Use the |Select View| button to choose a file to analyse from a list of already open .cfs files
' or press |New File| to load a  data file from disk.
'Use the Go to Frame command on the View menu to navigate to the frame of interest or use the arrow keys at the 
'bottom left hand corner of the frame.

'When the frame of interest is displayed, select the time range to convert by dragging the cursors.
'If the cursors are not visible you can bring them into view with the |Fetch Cursors| button. 
'Alternatively, press the |Whole Frame| button to convert the entire frame.

'Select the channels to be included in the .wav file by clicking on the channel numbers. (Hold down Cntrl to select multiple channels).
'Then press the |Export as .wav| button. A dialog appears with a drop-down menu to select 8-bit or 16-bit integer
'format. Checkboxes indicate the currently selected waveform channels. Ensure that the required channels are selected
'(maximum 4), then press OK. A File Save dialog then appears for entering the name and path of the output file.
'You must explicitly give the file a '.wav' filetype. Press Save to generate the output file.
'You can repeat this sequence with different frames and/or files as often as required creating a new .wav file each time. 
'Press |Quit| when you have finished to close the script.

'The script has a simple .wav file playback facility.
'You can play back the most recently saved .wav file via the computer's sound system provided that the file is relatively small (<1.5MB)
'and has no more than two channels. To do this, click the PlayBack button.  An error message is displayed if the .wav file is not suitable.
'Larger files can be replayed with a media player.
'The |StopPlay| button interrupts playback of a wav file.
 
'****************************************************************************

    const BSZ% := 8000;                    'limit array to the size that can be written quickly
    const swquit%:=1, swselvw%:=4, swnewf%:=5;'button numbers
    const swfetch%:=7,swall%:=8,swstore%:=10;
    const swplay%:=12,swpstop%:=13,on%:=1,off%:=0;
    var   fh%, bh%;
    var   st, fin;
    var   gFloat%[20];                          'global for floating window states
    var   views%[100];
	var   outfname$;							'name of most recent .wav file
    var   playflg%:=0;                          'flag that  wav file can be played back with Sound()
    var   nbufs%:=0;                            'number of 8K buffers
    var   list%[2],n%;
'****************************************************************************

    ToolbarClear();
    View(App(3)).WindowVisible(2);                          'iconise the script
    HideAll();
    if ViewKind(View(0)) = 0 then                           'if the current view is a cfs file use it as default 
       fh%:=View(0);
       WindowVisible(1);
    else                                                    'else find another open time view if available
        n%:=ViewList(list%[],1);
        if n% > 0 then
            fh%:=list%[1];
            View(fh%);
            WindowVisible(1);
        endif;
    endif;
    ToolbarSet(0,           "Idle",             Idle%);
    ToolbarSet(swquit%,     "&Quit",            Quit%);
    ToolbarSet(swselvw%,    "Select &View ",    Selview%);
    ToolbarSet(swnewf%,     "&New File",        OpenFile%);
    ToolbarSet(swfetch%,    "&Fetch Cursors",   FetchCursors%);
    ToolbarSet(swall%,      "&Whole Frame",     WholeFile%);
    ToolbarSet(swstore%,    "&Export as .Wav ", Storewav%);
	ToolbarSet(swplay%,     "&PlayLast ",       PlayWav%);
	ToolbarSet(swpstop%,    "&StopPlay ",       StopPlay%);
    ToolbarEnable(swplay%,off%);                                 'disable the playback buttons
    ToolbarEnable(swpstop%,off%);
    Toolbar(" Drag cursors to select time range to export", 191);
    halt;

'****************************************************************************

    func Idle%()
    var VL%,enable% := 0;
    var x%[2],nc%;							                       'temporary array to get size of chans%[]
    nc%:=ChanList(x%[], 1);
    var chans%[nc%+1];
    VL%:= Viewlist(views%[], 1);
    
    if VL% < 2 then                                             'disable view select button if only one available
        ToolbarEnable(swselvw%,off%); 
    else 
        ToolbarEnable(swselvw%,on%); 
    endif; 
    
    if vl% > 0 then                                             'enable most buttons if at least one  .cfs document open
      enable% := 1; 
    else 
        ToolbarText("");                                        'clear toolbartext if no file
    endif; 
    ToolbarEnable(swfetch%,  enable%);
    ToolbarEnable(swall%,  enable%);
    
    if ((ViewKind(View(0)) = 0) AND (View(0).ChanList(chans%[], 1) = 0)) then  'no waveform channels 
       enable% := 0; 
    endif; 
    ToolbarEnable(swstore%, enable%);
    
    if ViewKind(View(0)) = 0  then                                  'Renumber and relabel cursors if they are 'crossed'
        if Cursor(1) > Cursor(2) then ' 
            CursorRenumber(); 
            DoCursors%(); 
        endif; 
    endif;
    return 1;
    end;

'****************************************************************************

    func PlayWav%();
    if nbufs% < 200 then                                                'if the file is relatively small
	   Sound(outfname$,1);                                              ' play the wav file
    else
        Message("Sorry. Signal couldn't replay "+ outfname$ +"\n"+
        "Try using your media player.");
    endif;
        ToolbarEnable(swpstop%,on%);
	return 1;
	end;

'****************************************************************************

    func StopPlay%();
	Sound("S!",1);                                         'stop previous sound output and play system exclamation sound.
    ToolbarEnable(swpstop%,off%);
	return 1;
	end;

'****************************************************************************

    func FetchCursors%()									'Places cursors to be in middle of view
    var vh%; 
    View(fh%);
    CursorSet(2);
    DoCursors%(); 
    return 1;
    end;

'****************************************************************************

    func WholeFile%()									   'Places cursors to be in the start and end of the view
    View(fh%).CursorSet(2, 0, View(fh%).MaxTime());
    return 1;
    end;

'****************************************************************************

    func SetView%()                                       'Set this view current
    view(fh%);
	Window(0,0,100,100);
  	Draw(0, MaxTime());
    CursorSet(2);
    DoCursors%(); 
    ToolbarText(" Drag cursors to select time range to export"); 
	FrontView(fh%);
   return 1;
    end;

'****************************************************************************
    func DoCursors%(); 
    var vh%; 
    vh%:=View(); 
    view(fh%); 
    CursorLabel(4,1,"start(%5.2ps )"); 
    CursorLabel(4,2,"end(%5.2ps )"); 
    CursorLabelPos(1,1); 
    CursorLabelPos(2,3); 
    View(vh%); 
    return 1; 
    end;
'****************************************************************************
    func SelView%()
    var i%, ok%, n%;
    
    for i% := 0 to 99 do
        views%[i%] := 0;
    next;
    if (Viewlist(views%[], 1) > 0) then                 'Store viewhandles of all available time views
        var list$[views%[0]];
        for i%:=1 to views%[0] do
            view(views%[i%]);
            list$[i%-1] := windowtitle$();                'Set list of titles
        next;
        dlgcreate("Select view");                        'Define Menu ...
        dlglist(1,"Select view: ",list$[]);
        if (dlgshow(n%) = 0) then
            return 1;
        endif;
        fh%:=(views%[n%+1]);
        if (fh% > 0) then
            return SetView%();                             'Activate view
        endif;
    else 
        return OpenFile%();                              'Open new file
    endif;
    return 1;
    end;

'****************************************************************************

    func OpenFile%()
    var vh%;
    
    vh% := FileOpen("",0,0,"Select a .cfs file to convert");
    if (vh% < 0) then
       Sound("S*",1); 
       return 1;
    else
       fh% := vh%;
       return SetView%();
    endif;
    
    end;

'****************************************************************************

    func Storewav%()
    var   ok%, bytes%, fmt% := 1;
    var   x%[1];							                ' temporary array to get size of chans%[]
    var   sampInt;
    
    View(fh%);
    CursorRenumber();
    st  := Cursor(1);
    fin := Cursor(2);
    var chans%[ChanList(x%[],1)+1]; 
    
    ChanList(chans%[], 1);
    
    ok% := DoConfig%(chans%[], fmt%);                        ' Put up the configuration dialog
    if ok% <> 0 then                                         'dump if user selects a channel
        bh% := FileOpen("", 9, 1, "Filename for .wav file");
        outfname$:=Filename$();
        if bh% > 0 then
            ToolbarText("Saving .wav file"); 
            view(fh%);
            sampInt := BinSize(chans%[1]);
            docase
                case fmt% = 0 then bytes% := 1;
                case fmt% = 1 then bytes% := 2;
            endcase;
            SetupAndWriteHeader(bh%, sampInt, (fin-st)/sampInt, bytes%, chans%[0]);   'setup file header
            Dumpwave(bh%,  fh%, st, fin, chans%[0], bytes%, chans%[]);       '8/16-bit Integer 
            Sound("S!",1);
            View(bh%);
            FileClose();                                            'close the binary file
            if playflg% > 0 then
                ToolbarEnable(swplay%,on%);                         'enable the playback buttons
            endif;
            ToolbarText(" Drag cursors to select time range to export");'change prompt 
        else
            Sound("S*",1);
        endif ;
    endif;
    return 1;
    end

'****************************************************************************

    func DoConfig%(chans%[], &fmt%)
    var numchans%; numchans% := chans%[0];				    ' store number of channels
    var chansel%[numchans%]; 
    var ch%;
    var curchn% := 1;								        ' how many channels are selected?
    var ret%,									            ' return value from DlgShow to be returned
        nocurs% := 1,								        ' should we not use cursors to find st and fin
    	sep;										        ' which separator will we be using?
    var fmt$[4];
    var side% := 0, y%;
    var z; z := (numchans%+10.2)/2.0;
                   ' z is the y position (in DBUs) of the next line after the channel list.
                   ' z := (numchans%+2*row+1)/2.0; where row is the row the items to follow would be
                   ' on if we had no channel list.

                   ' Start new dialog, specifying w + h
    fmt$[0] := "8-bit Integer"; fmt$[1] := "16-bit Integer"; fmt$[2] := "32-bit IEEE real"; fmt$[3] := "64-bit IEEE real";

    DlgCreate ("Export in .Wav format",0,0,57,((numchans%+1)/2)+5); 
    DlgLabel  (1,"Output Format",2,1.3);
    DlgList   (2,13,fmt$[],2,18,1.3);							    ' create listbox
    DlgLabel  (3,"Choose the channels to export",2,3); 

	' We now loop through all the channels and create a checkbox for each. We arrange
	' them into two columns to decrease the height of the dialog box.

	for y% := 1 to numchans% do
		ch% := chans%[y%];
        chansel%[y%-1]:=ChanSelect(ch%);                           'check the box if channel was selected
		DlgCheck(y%+3,Print$("%3d %s %.2f KHz",ch%,ChanTitle$(ch%), 1.0/BinSize(ch%)/1000.0),25*side%+6,(y%+1)/2+3);
		side% := not side%;
	next;
    if ArrSum(chansel%[]) =0 then                                   'if no channels selected set 1st by default
        chansel%[0]:=1;
    endif;
	ret% := DlgShow(z,fmt%,z, chansel%[]);', z);                   ' ret% is 0 if user cancels
    if (ret% > 0) then
       	for y% := 1 to numchans% do					                ' count them and eliminate those
    	     if chansel%[y%-1] then						            ' which aren't
    		    chans%[curchn%] := chans%[y%];
    			 curchn% += 1;
    		  endif;
        next;
       	chans%[0] := curchn%-1;							              ' save number of selected channels

        if chans%[0] <=2 then
            playflg%:=1;                                               'flag that this file can be played back by the script
        else
            playflg%:=0;
        endif;                                                          'because it is a single channel
        if chans%[0] > 4 then 
            Message("Error|Can't export more than 4 channels.\n"+
            "Please go back and select 1 to 4 channels,"); 
            return 0; 
        endif; 
        if chans%[0] = 0 then							           ' validate data; must have
            Message("Error|You must select a channel to export!"); 
            return 0;									           ' at least one selected chan
        endif;
        if (chans%[0] > 1) then							           ' validate data; must have the same sampling rate
            for y% := 1 to chans%[0]-1 do					             ' count them and eliminate those
                if (View(fh%).BinSize(chans%[y%]) <> View(fh%).BinSize(chans%[y%+1])) then
                    Message("Error|You must select channels with the same sampling rate!");
                    return 0;									         ' at least one selected chan
                endif;
            next;
        endif;
       if st = fin then								               ' must have a nonzero period
            Message("Error|You must select a time region to export!"); 
            return 0;									               ' of time to dump
        endif;
    endif;
    return ret%;									                  ' return appropriate value
    end;

'****************************************************************************

    Proc SetupAndWriteHeader(bh%, sampInt, arrSiz%, bytes%, chans%);         ' Standard components of wav file header in integer format
    var   MyHead%[11];                     'array to store WAV format file header
       Myhead%[0]  := 0x46464952;                               ' RIFF
       MyHead%[1]  := 44 + (arrSiz%*bytes%*chans%);
       Myhead%[2]  := 0x45564157;                               ' WAVE
       Myhead%[3]  := 0x20746D66;                               ' fmt<sp>
       Myhead%[4]  := 0x10;                                     ' formatsize=16,
       Myhead%[5]  := 0x10000 * chans% + 0x01;                  ' upper 16 == chans, lower 16 == format tag == 1
       MyHead%[6]  := 1.0 / sampInt;                            ' Fill in the gaps in the header array
       MyHead%[7]  := MyHead%[6] * bytes% * chans%;             ' average bytes/second
       Myhead%[8]  := (0x80000 * bytes%) + (bytes%* chans%);    ' upper 16 == bits/sample, lowest 16 == block allignment
       Myhead%[9]  := 0x61746164;                               ' DATA
       MyHead%[10] := arrSiz% * bytes% * chans%;
    
       View(bh%).Bwrite(MyHead%[]);                             'write the header
    end;

'****************************************************************************
    proc DumpWave(bh%, fh%, sTime, eTime, nch%, bytes%, list%[]); '===modified from Spike 2 version GH 21/03/2003 
                                      
    var myArray%[BSZ%];
    var myArray[BSZ%]; 														'===real version of results array
    var scalefactor[4],lo,hi; 												'===yrange of channel
    var dat%[nch%][BSZ%];                                                   'generate channels array
    var n%, ich%, pts%;
    var startbin%; 															'===starting point of data
    
    View(fh%); 
    for iCh% := 0 to nCh%-1 do 
    	MinMax(list%[ich%+1],stime,etime,lo,hi); 
    	lo:=Abs(lo); 
    	hi:=Abs(hi); 
        scalefactor[ich%] := 1.0E-3; 
        if (Max(lo,hi) <> 0.0) then 
    	    scalefactor[ich%] := 32767.0/Max(lo,hi);  					 '===calculate suitable scale factors for each channel
        endif;  								                         '===calculate suitable scale factors for each channel
    next; 
    nbufs%:=0;                                                                  'initialise buffers counter    
    repeat
        for iCh% := 0 to nCh%-1 do
            View(fh%); 
      	 	n%:=ChanItems(list%[ich%+1],stime,etime);				             '===number of data points left to do
    		if n% > BSZ% then												     '===if > buffersize do one buffer's woth
                n%:=BSZ%
    		endif;
    		if n% > 0 then                                                       'if we got data
                startbin%:=XtoBin(list%[ich%+1],stime);
                ArrConst(MyArray[],View(fh%,list%[ich%+1]).[startbin%:n%]); 	'===copy real data to buffer
                ArrMul(MyArray[],scalefactor[ich%]); 						     '===scale it
                ArrConst(MyArray%[],MyArray[]); 								'===convert to integer
                pts% := n%;
                if (bytes% = 1) then
                    ArrDiv(myarray%[:n%], 256);
                    ArrAdd(myarray%[:n%], 128);                                 'using array arithmetic
                endif;
                ArrConst(dat%[ich%][:n%], myArray%[:n%]);                       'shift data
            endif;
            nbufs%+=1;
        next;
       View(bh%).BWriteSize(bytes%, dat%[][:pts%]);                                 'Output it
       stime := stime + pts% * BinSize(list%[ich%]);                                'time of next point
    until ((n% <= 0) or (stime >= etime));                                          'until no points left
    end;

'****************************************************************************

    func Quit%()                                                       'Quit button pressed
    View();
    RestoreAll();   
    return 0;
    end;

'****************************************************************************

    proc HideAll()
    var i%;
    gFloat%[0] := App(-2);                                              'number of windows
    for i% := 1 to gFloat%[0] do                                        'hide all windows and save state
       gFloat%[i%] := View(App(i%)).WindowVisible(0);
       next;
    end

'****************************************************************************

    proc RestoreAll()
    var i%;
    for i% := 1 to gFloat%[0] do                                        'restore hidden windows
       View(App(i%)).WindowVisible(gFloat%[i%]);
       next;
    end;

'****************************************************************************

